

ATARI®

Home Computer
ATARI

INVITO A PROGRAMMARE 3

Manuale d'Uso
Versione Cassetta

1. INTRODUZIONE

INVITO A PROGRAMMARE 3 è un corso il cui scopo consiste nel dare un'introduzione dei concetti di base per la programmazione sonora e grafica. INTRODUZIONE AL SUONO definisce i concetti di base di teoria musicale e le caratteristiche di un suono creato dal computer. INTRODUZIONE ALLA GRAFICA spiega l'uso dei colori nella grafica ed introduce la teoria dell'animazione. Poichè questo corso è il terzo di una serie, occorre conoscere il BASIC ATARI per poter fare un uso ottimale del presente corso. Per qualsiasi chiarificazione delle istruzioni di programmazione BASIC, occorre fare riferimento a INVITO A PROGRAMMARE 1 e 2 o al Manuale di Riferimento BASIC ATARI.

Il corso è formato da:

- Una cassetta Introduzione al Suono che contiene sia il programma che la parte audio del corso.
- Una cassetta Introduzione alla Grafica che contiene sia il programma che la parte audio del corso.
- Il presente manuale che contiene una parte da usarsi come guida di riferimento, e una parte di studio.

2. Spiegazione del Corso

ELEMENTI ATARI NECESSARI

- Un computer ATARI 800 XL
- Registratore ATARI 1010
- Cassetta Introduzione al Suono
- Cassetta Introduzione alla Grafica

ISTRUZIONI GENERALI PER IL CARICAMENTO DELLA CASSETTA

1. Collegare il computer all'apparecchio televisivo (fare riferimento al manuale operativo del computer).
2. Assicurarsi che il REGISTRATORE sia debitamente collegato al computer (consultare il manuale operativo del registratore).

NOTA: Se dovessero sorgere dei problemi inerenti il caricamento del programma e se oltre al registratore fossero connesse al computer altre unità periferiche, si consiglia di scollegare le altre periferiche e collegare il registratore direttamente al computer per eliminare più facilmente l'origine dell'inconveniente. Nell'eventualità che il funzionamento anomalo dovesse persistere e non fosse imputabile ad errori procedurali di caricamento, consultare il manuale operativo del registratore o del computer.

3. Accendere l'apparecchio televisivo (regolate il volume direttamente sull'apparecchio televisivo, poichè la traccia audio proviene dagli altoparlanti di quest'ultimo).
4. Attivare il computer ponendo in posizione **ON** l'interruttore POWER.
5. Se l'intero sistema è collegato appropriatamente, dovrebbe comparire sullo schermo del televisore la parola **READY** ed il cursore dovrebbe essere posto immediatamente sotto quest'ultima.
6. Premere **STOP/EJECT** sul REGISTRATORE per aprire lo sportello in cui viene inserita la cassetta.
7. Caricare la Lezione Uno inserendo la cassetta con la parte 1 rivolta verso l'alto e il nastro rivolto verso di voi.
8. Inserire la cassetta nell'apposito alloggiamento e richiudere lo sportello.
9. Qualora fosse necessario, premere **REWIND** per far riavvolgere il nastro all'inizio (fissare il contagiri sul registratore su 000).
10. Digitare **CLOAD** sulla tastiera del computer e premere **RETURN**. Un segnale acustico indicherà che occorre premere il tasto **PLAY** del registratore.

11. Premere il tasto **RETURN** per far partire il nastro. Attraverso la finestrella del registratore è possibile vedere il nastro che gira. I segnali acustici provenienti dall'apparecchio televisivo indicano che si sta caricando la Lezione Uno nella memoria del computer.
12. Quando viene visualizzato sullo schermo televisivo il messaggio **READY**, il primo programma sulla cassetta, **LEZIONE UNO**, è stato caricato nel computer. Assicurarsi che il tasto **PLAY** del registratore sia ancora premuto (e rimanga abbassato). Sull'apposito diagramma, segnare i giri indicati sul contagiri del registratore. Inserire tale cifra nella colonna **START AUDIO** della Lezione Uno. Qualora si voglia ritornare sulla Lezione, basta semplicemente riavvolgere il nastro fino ad arrivare al numero desiderato. Man mano che si procede con il corso, tenere nota delle cifre relative a **START AUDIO** di tutte le altre lezioni.
13. Digitare **RUN** e premere il tasto **RETURN** per iniziare la Lezione Uno. D'ora in poi, seguire le istruzioni che vengono enunciate nel corso della lezione. Alla fine della Lezione Uno, prendere nota del numero sul contagiri del registratore. Inserire tale numero nella colonna **END AUDIO** della Lezione Uno.
14. **Non** riavvolgere il nastro della cassetta. Quando si è pronti per iniziare la Lezione Due, seguire la stessa procedura applicata per la Lezione Uno (iniziare dal punto 10). Seguire la stessa procedura per tutte le lezioni del corso e per il test finale.
Per iniziare una lezione specifica, riavvolgere completamente la cassetta. Fissare il contagiri su 000. Premere il tasto **ADVANCE** del registratore. Far avanzare il nastro fino a che raggiunga il numero **START LOADING** di cui si è preso nota sul diagramma.

Preparazione per le lezioni pratiche

La memoria di un computer è suddivisa in **ROM** e **RAM**.

ROM. Read-Only Memory che contiene programmi che rimangono permanentemente nel computer e sono stati scritti dalla Casa Produttrice. Queste informazioni sono essenziali per il funzionamento del computer. I programmi ROM sono chiamati di **sola lettura** perchè possono solamente essere letti e non possono venire modificati in alcun modo.

RAM. Random Access Memory che contiene temporaneamente programmi e dati. È possibile inserire direttamente le informazioni nella RAM dalla tastiera; ciò permette la creazione di nuovi programmi, oppure di copiare dati dal dischetto o cassetta nella RAM.

Lo spegnimento del computer provoca la cancellazione di tutte le informazioni inserite nella RAM. Per poter inserire nuovi dati senza dover spegnere e riaccendere il computer è possibile usare l'istruzione **BASIC NEW** seguita da **RETURN**. Il comando **NEW** cancella tutte le informazioni presenti nella RAM.

Il comando **CLOAD** serve per trasferire dalla cassetta alla RAM un programma. Per cancellare questo programma e prepararsi ad esercitazioni pratiche, **OCCORRE** dare il comando **NEW**.

All'inizio di ogni esercitazione pratica verrà ripetuta la frase qui sotto riportata poiché è molto importante.

NOTA: Per eliminare quanto in memoria RAM, scrivere **NEW** e premere **RETURN** prima di ciascuna esercitazione pratica e fra ciascun programma esemplificativo.

ERRORI DI CARICAMENTO

Se durante il caricamento avete i messaggi d'errore 138 o 143 il "TAPE LEADER" (la parte iniziale del nastro magnetizzata) potrebbe essere troppo lungo. Il risultato è che il computer tenta di leggere dei dati dove non esistono. Riavvolgete il nastro all'inizio. Poi fate avanzare il nastro di circa 5 unità verificandolo sul contatore del registratore. Predisponete il registratore in **PLAY** e ripetete nuovamente la procedura di caricamento.

PERDITA DI SINCRONISMO

Potrebbe verificarsi che il parlato del nastro, causa impulsi esterni, perda il sincronismo con le immagini del computer. Interrompete l'esecuzione del programma con il tasto **RESET**. Digitare **POKE 54018,52** e premete il tasto **RETURN**. Questa istruzione vi consentirà, premendo il tasto **PLAY**, di riascoltare il nastro come su un normalissimo registratore. Con l'aiuto dei tasti **REWIND** ed **ADVANCE** del registratore cercate l'inizio del parlato della lezione in ascolto. Trovato il punto d'inizio premete il tasto **RESET** per interrompere la lettura del registratore. Poi digitate **RUN** per iniziare la lezione.

Alcune lezioni sono suddivise in PRIMA PARTE e SECONDA PARTE. La seconda parte della lezione sarà caricata automaticamente. Qualora si verificassero errori di caricamento, con il metodo sopraenunciato cercate la fine del parlato della prima parte della lezione; caricate poi la seconda parte della lezione con l'istruzione **CLOAD**.

Con questo metodo, oltre all'ausilio del diagramma promemoria inizio e fine lezioni, potrete ripetere qualsiasi lezione cercandola sul nastro e posizionando la cassetta sul punto dove lo Speaker informa della fine della lezione.

Per un migliore impiego del registratore è buona norma pulire periodicamente con i prodotti appositi la testina di lettura e di registrazione.

Per eventuali altri problemi consultate il Vostro Rivenditore ATARI di fiducia o il centro Assistenza ATARI.

COME RISPONDERE ALLE DOMANDE

Durante il corso vi verranno poste una serie di domande alle quali dovrete rispondere. Il formato della risposta si presenta in vari modi.

A) Lo speaker formula il quesito. Il video vi mostra la domanda e le relative risposte numerate (1, 2, 3). Rispondete con il numero relativo alla risposta che ritenete esatta (1, 2, 3 ecc.).

B) Come sopra.

Le risposte sono contraddistinte da lettere (A, B, C ecc.). Rispondete con la lettera relativa alla risposta che ritenete esatta (A, B, C ecc.).

C) Lo speaker formula il quesito. Il video vi mostra la domanda. La risposta deve essere per esteso.

Esempio:

SETCOLOR 1, 10, 8 - POSITION 2, 4 - DO-RE...

Fate molta attenzione quando rispondete a rispettare gli spazi nella risposta.

D) Domande con riferimento al manuale. Analizzate l'UNITÀ relativa del Manuale per poter rispondere. Il programma si arresterà per darvi tutto il tempo di ricerca.

3. Riassunto delle lezioni: Introduzione al suono

Lezione Uno: Registri Sonori

Vengono descritte le varie caratteristiche di suoni creati con il computer. Potete sperimentali modificando le varie caratteristiche ed ascoltando i suoni risultanti.

Lezione Due: Effetti Sonori

Vengono spiegati i principi basilari per la creazione di routine per effetti sonori semplici, come il tuono o l'esplosione. Si apprende anche come imitare gli strumenti musicali.

Lezione Tre: Routine Sonore

Viene effettuata la spiegazione, mediante fasi logiche, di come sviluppare una routine per creare musica computerizzata.

Lezione Quattro: Programmi Sonori Efficaci

Vengono introdotte le istruzioni DATA ed altre tecniche che chiariscono ulteriormente quanto spiegato nella Lezione Tre.

Lezione Cinque: ABC musicale

Intensità e rappresentazione dei suoni mediante note. Vengono descritti i pentagrammi e le chiavi.

Lezione Sei: Riconoscimento delle Note

Vengono suonati e visualizzati sul video i vari suoni musicali. Si riconoscono i suoni ed il loro valore sonoro. Alla fine della Lezione, dovreste trasporre la musica scritta nel codice del computer.

Lezione Sette: Routine Sonore con Vari Registri Sonori

Usando vari registri insieme, viene spiegato come creare programmi musicali ed effetti sonori.

Quiz Finale: INTRODUZIONE AL SUONO

Quiz riguardante tutto ciò che è stato spiegato con le lezioni del corso.

Figura 2: Tabella per il conteggio dei giri sul registratore relativa alla INTRODUZIONE AL SUONO.

INTRODUZIONE AL SUONO	LEZIONE	VALORI CONTAGIRI SUL REGISTRATORE		
		START LOADING	START AUDIO	END AUDIO
PARTE 1	1			
	2			
	3			
	4			
PARTE 2	5			
	6			
	7			
	TEST FINALE			

4. Descrizione delle unità per l'introduzione al suono

LEZIONE UNO: REGISTRI SONORI

UNITÀ 1: Il registro sonoro

Tre sono le variabili che determinano le caratteristiche del suono prodotto da un registro sonoro: nota o frequenza, distorsione e volume.

- La frequenza o nota, nel computer ATARI, varia da 1 a 255. Più grande è il Numero, minore sarà la frequenza (Nella unità 19 vengono enunciati i valori dei suoni musicali).
- La distorsione è una variabile che vi permette di creare effetti sonori speciali. I valori di distorsione vanno dallo 0 al 15. Solo con il valore 10 si avrà un suono puro, mentre tutti gli altri valori produrranno effetti particolari, specialmente con un numero di frequenza intorno al 200.
- L'ultima variabile serve per controllare il livello sonoro (volume) del suono. I valori vanno dall'1 al 15; il numero 1 produce un suono appena percettibile e il numero 15 produce un suono alto. Il numero 8 è considerato un volume "normale".

UNITÀ 2: I Quattro Registri Sonori

Il computer ATARI ha quattro registri sonori e sono numerati dallo 0 al 3.

UNITÀ 3: Istruzioni Sonore

Per produrre un suono, occorre inserire le quattro variabili nell'istruzione SOUND: il registro sonoro, il valore della frequenza, il valore della distorsione e il volume del suono. I valori devono essere separati dalla virgola.

ESEMPIO:

SOUND 0, 100, 10, 12 in cui: 0 è il registro sonoro, 100 è il valore della frequenza, 10 è il valore della distorsione e 12 è il valore del volume.

UNITÀ 4: Eliminazione del Suono

Regolate il volume dell'apparecchio televisivo in modo di udire distintamente i suoni; qualsiasi registro sonoro, una volta attivato, rimane attivo fino a che:

1. Venga assegnato un valore 0 al volume.
2. Venga programmata un'istruzione END, ad esempio:
10 SOUND 1, 50, 10, 8
20 FOR X = 1 TO 1000: NEXT X
30 END

3. Si preme il tasto **RESET** posto sul lato destro della tastiera.
(**BREAK** non interrompe il suono)
4. Venga disattivato il sonoro dell'apparecchio televisivo.

UNITÀ 5: Esercitazione pratica della Lezione Uno

NOTA: Per cancellare quanto contenuto nella memoria RAM del computer, scrivere **NEW** e premere il tasto **RETURN** prima di iniziare ogni parte pratica e fra un programma esemplificativo e l'altro.

1. Scrivete questa istruzione sonora nel computer; poi, premete il tasto **RETURN**
SOUND 0, 50, 2, 8
Per arrestare il suono, scrivere **END**, poi premere **RETURN**
2. Modificare il valore della frequenza nelle istruzioni enunciate in fase 1. Continuate a cambiare il valore fino a che non avete acquisito un'idea generale della gamma di valori disponibili. (Suggerimento: fare riferimento alla Unità 22).
3. Allo stesso modo, sperimentate i valori di distorsione e livello sonoro.

Non appena pronti caricate la lezione successiva, scrivendo **CLOAD** e premendo due volte il tasto **RETURN**. Non appena il messaggio **READY** viene visualizzato sullo schermo, scrivete **RUN** e premete il tasto **RETURN** per iniziare la lezione.

LEZIONE DUE: EFFETTI SONORI

UNITÀ 6: Riassunto della Lezione Due

1. Usate le variabili nelle istruzioni SOUND
SOUND 0, X, Y, Z
SOUND 0, frequenza, 10, 8
SOUND 0, 15, 2, VOLUME
2. Create effetti sonori con il loop **FOR/NEXT**
10 FOR FREQUENZA = 1 to 225
20 SOUND 0, FREQUENZA, 10, 8
30 NEXT FREQUENZA
3. Create una gamma di suoni variando il volume di un loop **FOR/NEXT**
10 REM VARIAZIONE DEL VOLUME
20 FOR VOLUME = 15 TO 0 STEP:1
30 SOUND 0, 15, 0, VOLUME
40 NEXT VOLUME
4. Suoni intermittenti, esplosioni, musica prodotta con colpi ecc. vengono tutti creati con rapide variazioni ed interruzioni.

UNITÀ 7: Esercitazione Pratica della Lezione Due

Nota: Per cancellare quanto contenuto nella memoria RAM del computer, scrivere **NEW** e premere il tasto **RETURN** prima di iniziare ogni parte pratica e fra un programma esemplificativo e l'altro.

Provate con questi programmi sonori

```
READY
10 REM TUONO
20 FOR TONO=1 TO 255
30 SOUND 0, TONO, 8, 15
40 NEXT TONO
```

```
READY
10 REM LOCOMOTIVA A VAPORE
20 FOR VOLUME= 15 TO 0 STEP -1
30 SOUND 0, 15, 0, VOLUME
40 NEXT VOLUME
50 GOTO 10
```

```
READY
10 REM SIRENA INTERMITTENTE
20 FOR TONO=100 TO 10 STEP -1
30 SOUND 0, TONO, 10, 8
40 NEXT TONO
50 FOR TONO= 10 TO 100
60 SOUND 0, TONO, 10, 8
70 NEXT TONO
80 GOTO 20
```


Per imitare l'organo si usano due loop di volume nel programma seguente. Il secondo loop arrotonda il suono proprio come accade quando colui che suona toglie il dito dal tasto.
La linea 10 del programma sceglie la nota casualmente.

```
READY
5 REM ORGANO A CANNE
10 TONO = INT (RND(0) * 255)
20 FOR VOLUME = 0 TO 15
30 SOUND 0, TONO, 10, VOLUME
35 NEXT VOLUME
40 FOR VOLUME = 15 TO 0 STEP -1
50 SOUND 0, TONO, 10, VOLUME
55 NEXT VOLUME
60 GOTO 10
```

```
READY
10 REM PIANOFORTE
20 TONO = INT(RND(0) * 155)
30 FOR VOLUME = 15 TO 0 STEP -1
40 SOUND 0, TONO, 10, VOLUME
50 NEXT VOLUME
60 GOTO 20
```

Aggiungete all'ultimo programma la linea seguente per suonare delle note simultaneamente: 45 SOUND 1, TONO +32, 10, VOLUME

Non appena pronti caricate la lezione successiva, scrivendo **CLOAD** e premendo due volte il tasto **RETURN**. Non appena il messaggio **READY** viene visualizzato sullo schermo, scrivete **RUN** e premete il tasto **RETURN** per iniziare la lezione.

LEZIONE TRE: ROUTINE SONORE

UNITÀ 8: Routine Sonora

Un programma che produce una serie di suoni viene definito routine sonora. Nelle unità dalla 9 alla 17 viene mostrata l'evoluzione di un metodo efficace di programmazione di routine sonora usando un unico registro sonoro. (Nella Lezione Sette parleremo dei programmi che hanno più di un registro sonoro).

UNITÀ 9: Numeri di linea

```
READY
100 SOUND 0, 81, 10, 8
110 SOUND 0, 64, 10, 8
120 SOUND 0, 53, 10, 8
130 SOUND 0, 64, 10, 8
140 SOUND 0, 60, 10, 8
150 SOUND 0, 47, 10, 8
160 SOUND 0, 60, 10, 8
170 END
```

UNITÀ 10: Aggiunta di istruzioni di ritardo

```
READY
100 SOUND 0, 81, 10, 8
105 FOR D=1 TO 300:NEXT D
110 SOUND 0, 64, 10, 8
115 FOR D=1 TO 300:NEXT D
120 SOUND 0, 53, 10, 8
125 FOR D=1 TO 300:NEXT D
130 SOUND 0, 64, 10, 8
135 FOR D=1 TO 300:NEXT D
140 SOUND 0, 60, 10, 8
145 FOR D=1 TO 300:NEXT D
150 SOUND 0, 47, 10, 8
155 FOR D=1 TO 300:NEXT D
160 SOUND 0, 60, 10, 8
165 FOR D=1 TO 300:NEXT D
170 END
```

Ogni singolo suono deve avere un Loop di ritardo

UNITÀ 11: Varie lunghezze delle note

```
READY
100 SOUND 0, 81, 10, 8
105 FOR D=1 TO 400:NEXT D
110 SOUND 0, 64, 10, 8
115 FOR D=1 TO 200:NEXT D
120 SOUND 0, 53, 10, 8
125 FOR D=1 TO 400:NEXT D
130 SOUND 0, 46, 10, 8
135 FOR D=1 TO 400:NEXT D
140 SOUND 0, 60, 10, 8
145 FOR D=1 TO 400:NEXT D
150 SOUND 0, 47, 10, 8
155 FOR D=1 TO 200:NEXT D
160 SOUND 0, 60, 10, 8
165 FOR D=1 TO 600:NEXT D
170 END
```

UNITÀ 12: Soubroutine di ritardo

```
READY
100 SOUND 0, 81, 10, 8
105 T=400:GOSUB 1000
110 SOUND 0, 64, 10, 8
115 T=200:GOSUB 1000
120 SOUND 0, 53, 10, 8
125 T=400:GOSUB 1000
130 SOUND 0, 64, 10, 8
135 T=400:GOSUB 1000
140 SOUND 0, 60, 10, 8
145 T=400:GOSUB 1000
150 SOUND 0, 47, 10, 8
155 T=200:GOSUB 1000
160 SOUND 0, 60, 10, 8
165 T=600:GOSUB 1000
170 END
1000 FOR D=1 TO T:NEXT D
1010 RETURN
```

UNITÀ 13: PEEK e POKE

Per poter effettuare dei conteggi accurati con le routine musicali, è possibile avvalersi del "contaminuti", o clock, posto all'interno della memoria del computer. Per localizzare l'ubicazione del contaminuti, occorre avere una istruzione **PEEK**. Per riportare il contaminuti a zero prima di usarlo nuovamente, occorre una istruzione **POKE**. La routine seguente sostituisce il loop FOR/NEXT che è meno preciso del contaminuti di computer.

```
READY
1000 POKE 20, 0:REM AZZERA
      IL CONTAMINUTI
1010 C = PEEK(20):REM IL CONTEGGIO
      DEL CONTAMINUTI VIENE ASSEGNATO
      ALLA VARIABILE C
1020 IF C < T THEN GOTO 1010:REM SE LA
      VARIABILE C È MINORE DI T VIENE
      LETTO NUOVAMENTE IL CONTAMINUTI
1030 RETURN:REM LA PAUSA VIENE INTERROTTA
      QUANDO C È UGUALE A T
```

UNITÀ 14: Esercitazione Pratica della Lezione Tre

NOTA: Per cancellare quanto contenuto nella memoria RAM del computer, scrivere **NEW** e premere il tasto **RETURN** prima di iniziare ogni parte pratica e fra un programma esemplificativo e l'altro.

Attivate il seguente programma per verificare il suono che viene emesso.

```
READY
100 SOUND 0, 81, 10, 8
110 SOUND 0, 64, 10, 8
120 SOUND 0, 53, 10, 8
130 SOUND 0, 64, 10, 8
140 SOUND 0, 60, 10, 8
150 SOUND 0, 47, 10, 8
160 SOUND 6, 67, 10, 8
170 END
```

Aggiungete le righe seguenti e attivate ancora il programma

```
READY
105 FOR D = 1 TO 400:NEXT D
115 FOR D = 1 TO 200:NEXT D
125 FOR D = 1 TO 400:NEXT D
135 FOR D = 1 TO 400:NEXT D
145 FOR D = 1 TO 400:NEXT D
155 FOR D = 1 TO 200:NEXT D
165 FOR D = 1 TO 600:NEXT D
```

Sostituite i loop **FOR/NEXT** con una subroutine Loop **FOR/NEXT** aggiungendo le righe seguenti al vostro programma. Attivate il programma ancora una volta.

```
READY
105 T = 400:GOSUB 1000
115 T = 200:GOSUB 1000
125 T = 400:GOSUB 1000
135 T = 400:GOSUB 1000
145 T = 400:GOSUB 1000
155 T = 200:GOSUB 1000
165 T = 600:GOSUB 1000
1000 FOR D = 1 TO T:NEXT D
1010 RETURN
```

Sostituite la subroutine **FOR/NEXT** con la subroutine seguente e attivate ancora il programma.

```
1000 POKE 20, 0
1010 C=PEEK(20)
1020 IF C < T THEN GOTO 1010
1030 IF C > =T THEN RETURN
```


NOTA: Con T = 400, la nota suona per circa sette secondi! Occorre diminuire i valori di T.

Sostituite la riga 1030 del programma qui sopra con la riga seguente e attivate ancora il programma. Vi è qualche differenza? Perché? 1030 RETURN

Non appena pronti caricate la lezione successiva, scrivendo **CLOAD** e premendo due volte il tasto **RETURN**. Non appena il messaggio **READY** viene visualizzato sulla schermo, scrivete **RUN** e premete il tasto **RETURN** per iniziare la lezione.

LEZIONE QUATTRO: PROGRAMMI SONORI

UNITÀ 15: Programma Musicale con Subrutine di temporizzazione



```
READY
100 PT=81:T=40:GOSUB 1000
110 PT=64:T=20:GOSUB 1000
120 PT=53:T=40:GOSUB 1000
130 PT=64:T=40:GOSUB 1000
140 PT=60:T=40:GOSUB 1000
150 PT=47:T=20:GOSUB 1000
160 PT=60:T=60:GOSUB 1000
170 END
1000 SOUND 0, PT, 10, 8
1010 POKE 20, 0
1020 C=PEEK(20)
1030 IF C < T THEN GOTO 1020
1040 RETURN
```

UNITÀ 16: Uso di istruzioni READ/DATA

```
READY
100 READ PT, T
110 IF PT=0 AND T=0 THEN END
120 GOSUB 1000
130 GOTO 100
1000 SOUND 0, PT, 10, 8
1010 POKE 20, 0
1020 C=PEEK(20)
1030 IF C < T THEN GOTO 1020
1040 RETURN
2000 DATA 81, 40
2010 DATA 64, 20
2020 DATA 53, 40
2030 DATA 64, 40
2040 DATA 47, 20
2050 DATA 60, 60
2060 DATA 0, 0
```

UNITÀ 17: Linee DATA scritte in modo efficiente

```
READY
2000 DATA 81, 40, 64, 20, 53, 40, 64, 40, 60, 40
2010 DATA 47, 20, 60, 60, 0, 0
```


UNITÀ 18: Esercitazione pratica della Lezione Quattro

NOTA: Per cancellare quanto contenuto nella memoria RAM del computer, scrivere **NEW** e premere il tasto **RETURN** prima di iniziare ogni parte pratica e fra un programma esemplificativo e l'altro.

Segue il nostro programma musicale standard. Le linee DATA contengono le note della Quinta Sinfonia di Beethoven. Attivate il programma così com'è scritto.

```
READY
100 READ PT, T
110 IF PT=0 AND T=0 THEN END
120 GOSUB 1000
130 GOTO 100
1000 SOUND 0, PT, 10, 8
1010 POKE 20, 0
1020 C=PEEK(20)
1030 IF C < T THEN GOTO 1020
1040 RETURN
2000 DATA 81, 30
2010 DATA 81, 30
2020 DATA 81, 30
2030 DATA 102, 90
2040 DATA 0, 0
```

C'è qualcosa che non va? Non vi sono pause fra le note! Ecco il rimedio più semplice: aggiungere questa riga al programma: 1040 SOUND 0, 0, 0, 0,; **RETURN**.

Ora ciascuna nota viene automaticamente interrotta prima che si senta la seguente. Qui di seguito vi sono enunciati i valori di intensità e di tempo della canzone "Twinkle Twinkle Little Star". Inserirli nel programma qui sopra come linee DATA. Assicuratevi che l'ultima linea DATA abbia un valore d'intensità 0 e un valore di pausa 0 per arrestare il programma.

PT	T
121	30
121	30
60	30
60	30
53	30
53	30
60	60

Ecco valori DATA di alcune canzoni che potreste inserire in un programma musicale.

READY
1999 REM CHRISTMAS THEME
2000 DATA 121, 10, 121, 10, 121, 20, 91, 10, 91,
10, 91, 20, 96, 10, 91, 10, 81, 10, 72, 10, 68,
10, 81, 10, 72, 30
2010 DATA 68, 10, 60, 20, 53, 10, 68, 10, 72, 10,
91, 10, 81, 20, 91, 60
2020 DATA 0,0

READY
1999 REM SAILOR'S WORK SONG
2000 DATA 85, 40, 85, 40, 72, 40, 72, 40, 108,
40, 108, 40, 144, 40, 108, 20, 85, 20, 72, 120
2010 DATA 64, 20, 85, 20, 72, 120, 72, 40, 85,
40, 85, 40, 72, 40, 72, 40, 108, 40, 108, 40, 144,
40, 108, 20, 96, 20
2020 DATA 85, 40, 72, 80, 85, 40, 96, 80, 108, 80
2030 DATA 0, 0

READY
1999 REM AMERICA THE BEAUTIFUL
2000 DATA 81, 30, 81, 45, 96, 15, 96, 30, 81, 30, 81, 45,
108, 15, 108, 30, 96, 30, 91, 30, 81, 30, 72, 30, 64,
30, 81, 90
2010 DATA 81, 30, 81, 45, 96, 15, 96, 30, 81, 30, 81, 45,
108, 15, 108, 30, 53, 30, 57, 30, 53, 30, 47, 30, 72, 30,
53, 90
2020 DATA 81, 30, 47, 45, 47, 15, 53, 30, 60, 30, 60, 45, 64,
15, 64, 30, 60, 30, 53, 30, 64, 30, 72, 30, 81, 30, 60, 90
2030 DATA 60, 30, 60, 45, 72, 15, 72, 30, 60, 30, 60, 45, 81,
15, 81, 30, 81, 30, 72, 30, 60, 30, 81, 30, 53, 30, 60, 90
2040 DATA 0, 0

È possibile usare il programma seguente per far in modo che i suoni qui sopra imitino la timbrica del pianoforte.


```
READY
10 REM PIANOFORTE
100 READ PT, T
110 IF PT=0 AND T=0 THEN END
120 GOSUB 1000
130 GOTO 100
1000 FOR VOLUME = 15 TO 0 STEP -0.5
1001 SOUND 0, PT, 10, VOLUME
1002 NEXT VOLUME
1010 POKE 20,0
1020 C = PEEK(20)
1030 IF C < T THEN GOTO 1020
1040 RETURN
2000 DATA 81, 30
2010 DATA 81, 30
2020 DATA 81, 30
2030 DATA 102, 90
2040 DATA 0, 0
```

Non appena pronti, caricate la Lezione Cinque. Occorre girare la Cassetta **INTRODUZIONE AL SUONO** sul LATO2. Consultate le istruzioni generali di caricamento della cassetta. Ricordate di azzerare il contagiri del registratore.

LEZIONE CINQUE: ABC MUSICALE


UNITÀ 19: Pentagrammi

Le note, rappresentate graficamente da simboli, determinano la frequenza dei suoni musicali. Le note vengono messe sul pentagramma che è formato da cinque linee e quattro spazi. Ciascuna linea e ciascuno spazio rappresentano una frequenza diversa. I due pentagrammi più usati sono quello in chiave di violino e quello in chiave di basso. All'inizio di ciascun pentagramma vi è sempre la chiave che lo identifica. Tre note: DO RE SI stanno fra un pentagramma e l'altro.



UNITÀ 20: Ulteriori Informazioni

Ecco quali sono le note posizionate negli spazi del pentagramma:



Ecco come sono posizionate le altre note sulle linee del pentagramma:




Non appena pronti caricate la lezione successiva, scrivendo **CLOAD** e premendo due volte il tasto **RETURN**. Non appena il messaggio **READY** viene visualizzato sullo schermo, scrivete **RUN** e premete il tasto **RETURN** per iniziare la lezione.


LEZIONE SEI: RICONOSCIMENTO DELLE NOTE

UNITÀ 21: Linee supplementari

Le linee supplementari sono dei trattini che vengono usati per continuare a scrivere le note al di sopra e al di sotto del pentagramma. Ecco alcune linee supplementari che possono essere impiegate per estendere il pentagramma in chiave di violino.



Ecco alcune linee supplementari al di sopra e al di sotto del pentagramma in chiave di basso.



UNITÀ 22: Valori ATARI delle note musicali.

NOTE ALTE	DO	29
	SI	31
	LA # o SI b	33
	LA	35
	SOL # o LA b	37
	SOL	40
	FA # o SOL b	42
	FA	45
	MI	47
	RE # o MI b	50
	RE	53
	DO # o RE b	57
	DO	60
	SI	64
	LA # o SI b	68
	LA	72
	SOL # o LA b	76
	SOL	81
Note di mezzo	FA # o SOL b	85
	FA	91
	MI	96
	RE # o MI b	102
	RE	108
	DO # o RE b	114
DO DI MEZZO	DO	121
	SI	128
	LA # o SI b	136
	LA	144
	SOL # o LA b	153
	SOL	162
	FA # o SOL b	173
	FA	182
NOTE BASSE	MI	193
	RE # o MI b	204
	RE	217
	DO # o RE b	230
	DO	243

UNITÀ 23: Esercitazione Pratica della Lezione sei

NOTA: Per cancellare quanto contenuto nella memoria RAM del computer, scrivere **NEW** e premere il tasto **RETURN** prima di iniziare ogni parte pratica e fra un programma esemplificativo e l'altro.

Il seguente programma musicale creato nel corso delle Lezioni precedenti, usatelo per trasporre le canzoni seguenti.

```
READY
100 READ PT, T
110 IF PT=0 AND T=0 THEN END
120 GOSUB 1000
130 GOTO 100
1000 SOUND 0, PT, 10, 8
1010 POKE 20, 0
1020 C = PEEK(20)
1030 IF C < T THEN GOTO 1020
1040 SOUND 0, 0, 0, 0:RETURN
2000 DATA .....
```

Di seguito illustriamo il simbolo di durata delle note con il relativo valore ATARI. Usate tali valori di durata e trasponete la musica dei pentagrammi in valori del computer in modo che possa essere eseguita.

 = 120

 = 60

 = 30

 = 15

BILLY BOY

Allegro



GO TELL AUNT RHODY

Adagio



LOVE SOMEBODY

Allegro



Non appena pronti caricate la lezione successiva, scrivendo **CLOAD** e premendo due volte il tasto **RETURN**. Non appena il messaggio **READY** viene visualizzato sullo schermo, scrivete **RUN** e premete il tasto **RETURN** per iniziare la lezione.

LEZIONE SETTE: ROUTINE SONORE CON REGISTRI MULTIPLI

UNITÀ 24: Routine Musicale e Nota singola

```
READY
100 READ PT, T
110 IF PT = 0 AND T = 0 THEN END
120 GOSUB 1000
130 GOTO 100
1000 SOUND 0, PT, 10, 8
1010 POKE 20, 0
1020 C = PEEK(20)
1030 IF C < T THEN GOTO 1020
1040 SOUND 0, 0, 0, 0: RETURN
2000 DATA 81, 40
2010 DATA 64, 20
2020 DATA 53, 40
2030 DATA 64, 50
2040 DATA 60, 40
2050 DATA 47, 20
2060 DATA 60, 60
2070 DATA 0, 0
```

UNITÀ 25: Riassunto della Lezione Sette

Gli effetti sonori vengono prodotti attivando o disattivando due o più registri sonori. Tuttavia, perchè parecchi registri possano funzionare insieme per produrre effetti utili, tali registri devono essere compatibili o "inserirsi". Ciò di solito implica che i registri sonori vengano programmati con toni puri.

Le note musicali sono compatibili o si "inseriscono" con altre note a seconda del rapporto matematico della frequenza delle note. Le note che sono troppo ravvicinate come valore numerico di suono tendono ad essere discordanti.

Un accordo musicale è formato da due o più note, generate da più registri, che vengono suonate simultaneamente.

UNITÀ 26: Esercitazione pratica della Lezione sette

NOTA: Per cancellare quanto contenuto nella memoria RAM del computer, scrivere **NEW** e premere il tasto **RETURN** prima di iniziare ogni parte pratica e fra un programma esemplificativo e l'altro.

Provate con il seguente programma di effetti sonori. Genera due suoni contrastanti che creano un effetto sonoro.

```
READY
5 REM SIRENA DI TRENO
10 FOR SUONO= 1 TO 2
15 SOUND 0, 190, 10, 8:SOUND 1, 175, 10, 8
20 FOR RITARDO= 1 TO 750:NEXT RITARDO
30 SOUND 0, 0, 0, 0: SOUND 1, 0, 0, 0
40 FOR RITARDO= 1 TO 50:NEXT RITARDO
60 NEXT SUONO
```

Ecco un altro programma che genera un effetto diverso:

```
READY
5 REM SIRENA ALLARME AEREO
10 FOR TONO= 100 TO 200
20 SOUND 0, TONO, 10, 8
30 SOUND 1, 300-TONO, 0, 0
40 FOR RITARDO= 1 TO 5:NEXT RITARDO
50 NEXT TONO
60 GOTO 10
```

Segue un programma con accordi musicali. Inserirlo e ascoltate i suoni generati. Notate che quando vengono attivati tutti i registri sonori il suono è migliore se il volume di ciascuno non supera il valore 8.

```
READY
100 READ PTO, PT1, PT2, PT3, T
110 IF PTO=0 AND PT1=0 AND PT2=0 AND PT3=0 AND T=0
THEN END
120 GOSUB 1000
130 GOTO 100
1000 SOUND 0, PTO,10,8:SOUND 1, PT1,10,8:SOUND 3,PT3,10,8
1010 POKE 20,0
1020 C= PEEK(20)
1030 IF C < T THEN GOTO 1020
1040 SOUND 0,0,0,0: SOUND 1,0,0,0:SOUND 2,0,0,0:SOUND
3,0,0,0:RETURN
2000 DATA 121, 96, 81, 60, 20
2010 DATA 162, 121, 96, 81, 20
2020 DATA 172, 121, 96, 81, 40
2030 DATA 162, 121, 96, 81, 40
2040 DATA 108, 92, 82, 64, 20
2050 DATA 162, 121, 96, 60, 100
2060 DATA 0, 0, 0, 0, 0
```

Inserite i seguenti dati nel programma precedente per ascoltare una musica che conoscete già.

```
READY
2000 DATA 243, 193, 162, 60, 20
2010 DATA 243, 193, 162, 60, 20
2020 DATA 243, 193, 162, 60, 20
2030 DATA 243, 193, 162, 81, 20
2040 DATA 182, 144, 121, 72, 20
2050 DATA 182, 144, 121, 72, 20
2060 DATA 243, 193, 162, 81, 40
2070 DATA 0, 0, 0, 47, 20
2080 DATA 0, 0, 0, 47, 20
2090 DATA 0, 128, 108, 53, 20
2100 DATA 0, 128, 108, 53, 20
3000 DATA 0, 0, 0, 0, 0
```

Il programma che segue è simile all'ultimo visto, ad eccezione del fatto che imita la timbrica del pianoforte. Potete risparmiare tempo usando quest'ultimo programma per la maggior parte delle linee e sostituendo solo quelle linee che sono diverse.

```

READY
10 REM PIANOFORTE
100 READ PTO, PT1, PT2, PT3, T
110 IF PTO=0 AND PT1=0 AND PT2=0 AND PT3=0 AND T=0
THEN END
120 GOSUB 1000
130 GOTO 100
1000 FOR VOLUME=8 TO 0 STEP -0,5
1002 SOUND 0, PTO, 10, VOLUME:SOUND 1, PT1, 10,
VOLUME:SOUND 2, PT2, 10, VOLUME:SOUND 3, PT3, 10, VOLUME
1005 NEXT VOLUME
1010 POKE 20,0
1020 C=PEEK(20)
1030 IF C < T THEN GOTO 1020
1040 RETURN
2000 DATA 243, 193, 162, 60, 20
2010 DATA 243, 193, 162, 60, 20
2020 DATA 243, 193, 162, 81, 20
2030 DATA 243, 193, 162, 81, 20
2040 DATA 182, 144, 121, 72, 20
2050 DATA 182, 144, 121, 72, 20
2060 DATA 243, 193, 162, 81, 40
2070 DATA 0, 0, 0, 47, 20
2080 DATA 0, 0, 0, 47, 20
2090 DATA 0, 128, 108, 53, 20
2100 DATA 0, 128, 108, 53, 20
2110 DATA 243, 193, 162, 60, 80
3000 DATA 0, 0, 0, 0, 0

```

Questo programma imita la timbrica di un organo.

```

READY
10 REM ORGANO
100 READ PTO, PT1, PT2, PT3, T
110 IF PTO=0 AND PT1=0 AND PT2=0 AND PT3=0 AND T=0
THEN END
120 GOSUB 1000
130 GOTO 100
1000 FOR VOLUME=0 TO 8
1002 SOUND 0, PTO, 10, VOLUME:SOUND 1, PT1, 10,
VOLUME:SOUND 2, PT2, 10, VOLUME:SOUND 3,P T3, 10, VOLUME
1005 NEXT VOLUME
1010 POKE 20,0
1020 C=PEEK(20)
1030 IF C < T THEN GOTO 1020
1036 FOR VOLUME=8 TO 0 STEP -0,5
1037 SOUND 0, PTO, 10, VOLUME:SOUND 1, PT1, 10,
VOLUME:SOUND 2, PT2,10, VOLUME:SOUND 3,P T3, 10, VOLUME
1038 NEXT VOLUME
1040 RETURN
2000 DATA 243, 193, 162, 60, 20
2010 DATA 243, 193, 162, 60, 20
2020 DATA 243, 193, 162, 60, 20
2030 DATA 243, 193, 162, 81, 20
2040 DATA 182, 144, 121, 72, 20
2050 DATA 182, 144, 121, 72, 40
2060 DATA 243, 193, 162, 81, 40
2070 DATA 0, 0, 0, 47, 20
2080 DATA 0, 0, 0, 47, 20
2090 DATA 0, 128, 108, 53, 20
2100 DATA 0, 128, 108, 53, 20
2110 DATA 243, 193, 162, 60, 80
3000 DATA 0, 0, 0, 0, 0

```

Complimenti! Continuate con il **quiz finale** dell'introduzione al SUONO. In questa parte non vi è più alcun audio.

UNITÀ 27: Programma da utilizzare per il test finale.

```
READY
100 RED PTO, PT1, PT2, PT3, T
110 IF PTO=0 AND PT1=0 AND PT2=0 AND PT3=0 AND T=0
THEN END
120 GOSUB 1000
130 GOTO 100
1000 SOUND 0, PTO, 10, 8:SOUND 1, PT1, 10, 8:SOUND 2, PT2, 10,
8:SOUND 3, PT3, 10,
1010 POKE 20,0
1020 C=PEEK(20)
1030 IF C < T THEN GOTO 1020
1040 SOUND 0, 0, 0, 0:SOUND 1, 0, 0, 0:SOUND 2, 0, 0, 0:
SOUND 3, 0, 0, 0:RETURN
2000 DATA 121, 96, 81, 20
2010 DATA 162, 121, 96, 81, 20
2020 DATA 172, 121, 96, 81, 40
2030 DATA 162, 92, 82, 64, 20
2040 DATA 108, 92, 82, 64, 20
2050 DATA 162, 121, 96, 60, 100
2060 DATA 0, 0, 0, 0, 0
```

5. Riassunto delle lezioni: Introduzione alla Grafica

Lezione Uno: Dimostrazione Grafica

Breve visualizzazione degli effetti più semplici ottenuti con la Grafica.

Lezione Due: Uso dei Colori

Istruzioni del codice colore e dell'uso dei registri colore.

Lezione Tre: Modi di Funzionamento Grafico 1 e 2

Introduzione ai modi di funzionamento grafico e spiegazione approfondita dei registri di colore.

Lezione Quattro: Grafica

Uso delle istruzioni PLOT e DRAWTO e dei modi di funzionamento grafico con colori.

Lezione Cinque: Trucchi con i modi di funzionamento grafico

Come usare l'intero video in qualsiasi modo grafico e passare da un modo a un altro senza eliminare la memoria dello schermo.

Lezione Sei: Posizione ed Animazione

Come posizionare i testi sullo schermo e dare l'impressione del movimento.

Lezione Sette: Caratteri Grafici

Come comporre immagini con simboli speciali.

Quiz Finale: Introduzione alla Grafica

Esame con quanto è stato trattato in questo corso.

Figura 3: Tabella per il conteggio dei giri sul contagiri del Registratore relativa all'Introduzione alla GRAFICA.

INTRODUZIONE ALLA GRAFICA	LEZIONE	VALORI CONTAGIRI SUL REGISTRATORE		
		START LOADING	START AUDIO	END AUDIO
PARTE 1	1			
	2			
	3			
	4			
PARTE 2	5			
	6			
	7			
	TEST FINALE			

6. Descrizione dell'Unità dell'Introduzione alla Grafica

LEZIONE UNO: DIMOSTRAZIONE GRAFICA

Non appena pronti, caricare la Lezione Due battendo **CLOAD** e premendo il tasto **RETURN** due volte. Dopo che è apparso sullo schermo il messaggio **READY**, battere **RUN** e premere **RETURN** per iniziare la Lezione.

LEZIONE DUE: USO DEI COLORI

UNITÀ 1: Tabella dei Colori del BASIC ATARI

Fare riferimento alla tabella seguente per trovare il numero assegnato al colore visualizzato sullo schermo.

Numero Colore	Colore sullo schermo
0	grigio
1	oro
2	arancio
3	arancio-rosso
4	rosa
5	porpora o violetto
6	blu-medio
7	blu
8	blu
9	blu chiaro
10	turchese
11	verde-mare
12	verde
13	verde-giallo
14	verde-arancio
15	arancio chiaro

NOTA: I colori possono variare con il tipo e la regolazione del televisore.

UNITÀ 2: Registri Colore

Il computer ATARI ha cinque registri di colore che vanno dallo 0 al 4. Ciascun registro colore ha un'applicazione specifica; per esempio, il registro colore 4 modifica il colore del bordo dello schermo ed il registro colore 2 modifica il colore di fondo dello schermo. Due numeri, il numero del colore e il corrispondente valore di Luminosità vengono associati a ciascun registro colore.

UNITÀ 3: SETCOLOR

SETCOLOR viene impiegato per assegnare colori specifici e relative intensità a ciascun registro colore. Ecco la sintassi di SETCOLOR: SETCOLOR X, Y, Z; dove X rappresenta il registro colore, Y rappresenta il numero colore e Z rappresenta il numero di luminosità.

UNITÀ 4: Ulteriori Informazioni sul SETCOLOR

La luminosità viene modificata ogni numero pari: 0, 2, 4, 6, 8, 10, 12 e 14.

NOTA: Il numero di luminosità 10 fa scomparire le scritte (o il testo). Per evitare tale situazione, riassegnare un numero di luminosità diverso e premere il tasto **RETURN**

Con le istruzioni SETCOLOR è possibile sostituire i numeri con delle variabili. L'unità seguente vi dà degli esempi pratici.

UNITÀ 5: Parte Pratica della Lezione Due

NOTA: Per cancellare quanto contenuto nella memoria RAM del computer, scrivere **NEW** e premere il tasto **RETURN** prima di iniziare ogni parte pratica e fra un programma esemplificativo e l'altro.

Provate a fare degli esercizi pratici usando SETCOLOR con il registro colore 2 o 4. Usate il modo di funzionamento diretto o il programma seguente per visualizzare tutti i colori e le intensità possibili sul vostro computer ATARI.

```
READY
10 REM PROGRAMMA PER VARIARE
  1 COLORI DELLO SCHERMO
20 REM USO DEL REGISTRO 2
30 FOR X=0 TO 15
40 FOR Y=0 TO 14
50 SETCOLOR 2, X, Y
55 FOR I=0 TO 100:NEXT I
60 NEXT Y
70 NEXT X
80 END
```

Sperimentate SETCOLOR sul Registro 1 per determinare l'intensità delle lettere sul video. Se le lettere scompaiono quando siete in modo di funzionamento diretto premere il tasto **RESET**.

```

READY
10 REM RAZZO
20 SETCOLOR 2, 0, 0
30 FOR CONTO = 10 TO 0 STEP -1
40 PRINT CONTO
50 FOR TEMPO = 1 TO 100:NEXT TEMPO
60 NEXT CONTO
65 SETCOLOR 2, 4, 14
70 PRINT CHR$(253); "PARTITO"
80 SETCOLOR 2, 0, 2
90 END

```

Non appena pronti caricate la lezione successiva, scrivendo **CLOAD** e premendo due volte il tasto **RETURN**. Non appena il messaggio **READY** viene visualizzato sullo schermo, scrivete **RUN** e premete il tasto **RETURN** per iniziare la lezione.

LEZIONE TRE: MODI DI FUNZIONAMENTO GRAFICO 1 E 2

UNITÀ 6: Modi Grafici

ISTRUZIONI COMPUTER	OGGETTO SUL VIDEO	DIMENSIONI (n. oggetti sul video)
GRAPHICS 0	carattere normale	24 righe di 40 caratteri
GRAPHICS 1	carattere grande (doppia larghezza)	20 righe di 20 caratteri
GRAPHICS 2	carattere grande (doppia larghezza e doppia altezza)	10 righe di 20 caratteri
GRAPHICS 3	quadrati grandi	20 righe di 40 quadrati
GRAPHICS 4	puntini	40 righe di 80 quadrati
GRAPHICS 5	puntini (a più colori)	40 righe di 80 quadrati

UNITÀ 7: Colore nel Modo di Funzionamento Grafico 1

I registri colore 2 e 4 rendono possibile la scelta del colore nel Modo di Funzionamento Grafico 1. Il registro colore 2 colorava la finestra del testo che viene visualizzata in basso sullo schermo.

Il registro colore 4 colora la finestra grafica. Utilizzate il comando SETCOLOR per immettere il colore e l'intensità che avete scelto nel registro colore appropriato.

SELECTOR 4, 10, 8

Il registro di colore 4
colora qui



SETCOLOR 2, 2, 2
La finestra riservata
al testo è colorata
dal registro 2

UNITÀ 8: Visualizzazione sul video nel modo di funzionamento Grafico 1 e 2.

Le normali istruzioni **PRINT** agiscono sulla finestra per il testo.

Per poter scrivere anche nella parte grafica dello schermo occorre usare un comando **PRINT** specifico. La sintassi è:

PRINT # 6; "le parole da visualizzare fra virgolette".

PRINT # 6; "Testo largo"



PRINT "Testo normale"

UNITÀ 9: Colorazione del testo nella finestra grafica

Perchè i caratteri del comando **PRINT # 6** siano visualizzati in determinati colori sullo schermo, programmate i registri di colore e il tipo di carattere.

AD ESEMPIO:

SETCOLOR 0, 9, 10 fissa il colore corrispondente al 9, blu chiaro, nel registro colore 0.

PRINT # 6: "TEST" farà apparire la parola **TEST** in blu chiaro nella finestra grafica.

Registro	Colore	Carattere usato	Istruzioni di esempio colore
0		Lettere maiuscole	SETCOLOR 0, 9, 10
1		Lettere minuscole	SETCOLOR 1, 15, 0
2		Maiuscole in reverse	SETCOLOR 2, 8, 6
3		Minuscole in reverse	SETCOLOR 3, 5, 14
4		Nessun carattere, solo sfondo	SETCOLOR 4, 5, 14

UNITÀ 10: Colori Preprogrammati

Se non viene dato alcun comando SETCOLOR prima delle istruzioni **PRINT # 6** nel programma, i colori dei caratteri appariranno in un colore preprogrammato nel registro colore.

Registro colore	Carattere usato	Colore preprogrammato
0	Lettere maiuscole	arancio
1	Lettere minuscole	verde chiaro
2	Maiuscole in reverse	blu scuro
3	Minuscole in reverse	rosso
4	Sfondo schermo	grigio scuro (nero)

UNITÀ 11: Parte Pratica della Lezione Tre

NOTA: Per cancellare quanto contenuto nella memoria RAM del computer, battere **NEW** e premere il tasto **RETURN** prima di iniziare ogni parte pratica e fra un programma esemplificativo e l'altro.

Inserire il modo di funzionamento Grafico battendo **GRAPHICS 1** e premendo il tasto **RETURN**. Battete diverse istruzioni **PRINT # 6**; per vedere che cosa accade sul video. Ripetete gli stessi con **GRAPHICS 2**.

NOTA: Le lettere in reverse vengono prodotte premendo per prima cosa il tasto bianco e nero sulla parte bassa della tastiera a destra. Per tornare ai caratteri normali, premere il tasto una seconda volta. Per ritornare al modo grafico normale, battere **GRAPHICS 0** e premere **RETURN**.

UNITÀ 12: Esercizi Pratici con SETCOLOR nel modo di funzionamento grafico 1 e 2.

Fate partire il seguente programma.

```
10 GRAPHICS 2
20 PRINT # 6; "UNO due TRE QUATTRO"
100 END
```

Notate il colore delle lettere sul video. Controllate la tabella dei colori e la tabella dei valori preprogrammati di colore. Ora, battete i seguenti comandi SETCOLOR dopo che l'esecuzione del programma è terminata.

Fate ciò quando siete ancora nel modo di funzionamento grafico usando i comandi in modo diretto (senza numero di linea):

SETCOLOR 0, 8, 0	osservate ciò che accade a "UNO"
SETCOLOR 1, 2, 10	osservate ciò che accade a "DUE"
SETCOLOR 2, 15, 14	osservate ciò che accade a "TRE"
SETCOLOR 3, 0, 6	osservate ciò che accade a "QUATTRO"

SETCOLOR 4, 4, 2 osservate il colore dello sfondo

Aggiungete al programma precedente le seguenti linee e fatelo eseguire ancora.

```
30 FOR X=0 TO 3: REM NUMERO DEI REGISTRI DI COLORE
40 FOR Y=1 TO 10: REM NUMERO DI LAMPI
50 SETCOLOR X, 0, 0
60 FOR TEMPO=1 TO 100:NEXT TEMPO:REM RITARDO
70 SETCOLOR X, 0, 14
80 FOR TEMPO=1 TO 100:NEXT TEMPO:REM RITARDO
90 NEXT Y
95 NEXT X
```


Premete il tasto **RESET**. Aggiungere la linea seguente e fatelo partire nuovamente:
15 SETCOLOR 4, 4, 10

NOTA: Ogniqualvolta viene eseguito il comando Graphics 0, 1 o 2 lo schermo viene pulito e vengono assegnati ai registri di colore i colori preprogrammati.

L'abbreviazione di GRAPHICS è GR.: quindi, Graphics 1 può essere scritto come GR.1 Non appena pronti caricate la lezione successiva, scrivendo **CLOAD** e premendo due volte il tasto **RETURN**. Non appena il messaggio **READY** viene visualizzato sullo schermo, scrivete **RUN** e premete il tasto **RETURN** per iniziare la lezione.


LEZIONE QUATTRO: GRAFICA

UNITÀ 13: Video del Modo di Funzionamento Grafico 3



UNITÀ 14: Comandi PLOT e DRAWTO

Per usare il comando **PLOT** immaginate una griglia sul video. Il comando **PLOT** serve per posizionare un punto all'intersezione delle coordinate X e Y, date mediante il comando **PLOT**. Le coordinate vanno dallo 0 al 39 sull'asse delle X e da 0 a 19 sull'asse dell'Y.



UNITÀ 15: Programma d'Esempio

Ricordate che le istruzioni **COLOR** non posizionano alcun colore nei registri. L'istruzione **COLOR** serve per INDICARE al computer quale registro controlla il colore del punto.

Determinate quale istruzione manca nel programma seguente:

```
10 GRAPHICS 3
20 SETCOLOR 3, 9, 4
30 PLOT 15, 6
```

UNITÀ 16: Comando COLOR e REGISTRI COLORE

Il comando **COLOR** controlla direttamente la scelta del registro di colore usato poi da qualsiasi comando **PLOT** o **DRAWTO** che lo segua. La tavola qui sotto illustra tale rapporto:

COLOR 1	corrispondente al registro di colore 0
COLOR 2	corrispondente al registro di colore 1
COLOR 3	corrispondente al registro di colore 2
COLOR 4	corrispondente al registro di colore 3

UNITÀ 17: Programma con il Comando COLOR

```
10 GRAPHICS 3
20 SETCOLOR 0, 7, 8
30 SETCOLOR 2, 0, 6
40 COLOR 1
50 PLOT 15, 15
```

Ricordate che il numero di colore 0 = grigio e il 7 = blu.

UNITÀ 18: Esercitazione Pratica della Lezione Quattro con l'istruzione COLOR

NOTA: Per cancellare quanto contenuto nella memoria RAM del computer, scrivere **NEW** e premere il tasto **RETURN** prima di iniziare ogni parte pratica e fra un programma esemplificativo e l'altro.

Fare partire il seguente programma:

```
10 GRAPHICS 3
20 SETCOLOR 0, 4, 10
30 COLOR 1
40 PLOT 10, 10
```

Cambiate i valori delle istruzioni COLOR e SETCOLOR, in modo da ottenere sul video dei quadratini con colori diversi. Cancellate la memoria **RAM** ed eseguite il programma seguente.

```
10 GRAPHICS 3
20 COLOR 1
30 X = INT(RND(0) * 40):Y = INT(RND(0) * 20)
40 PLOT X, Y
50 GOTO 20
```

Per arrestare la visualizzazione grafica casuale, premere il tasto **BREAK**. Aggiungere le seguenti linee al programma;

```
20 C = INT(RND(0)*4)
25 COLOR C
```

UNITÀ 19: Esercitazione Pratica per la Lezione Quattro con **PLOT** e **DRAWTO**. Batte e ed eseguite il seguente programma:

```
10 REM LAMPO LUMINOSO
20 GRAPHICS 3
30 COLOR 1
40 PLOT 10,0
42 DRAWTO 20, 19
```




Aggiungere le seguenti linee:

```
45 FOR X = 1 TO 10
50 SETCOLOR 0, X, 8
60 FOR D = 1 TO 100:NEXT D:REM RITARDO
70 SETCOLOR 0, 9, 14
80 FOR D = 1 TO 100:NEXT D
100 NEXT X
```

Non appena pronti, caricate la Lezione Cinque. Occorre girare la cassetta sul lato 2. Consultate le istruzioni generali per il caricamento della cassetta. Ricordate di azzerare il contagiri del registratore.

LEZIONE CINQUE: TRUCCHI CON I MODI DI FUNZIONAMENTO GRAFICO


UNITÀ 20: Disposizione Video con i Modi di Funzionamento Grafico 3, 4 e 5.

MODO GRAFICO	VIDEO	REGISTRI COLORE	MEMORIA
3		39 Disponibile per i punti grafici: 0, 1, 2. Sfondo determinato dal 4.	273 byte
4		79 Disponibile per i punti grafici: 0, 1, 2. Sfondo determinato dal 4	537 byte
5		79 Disponibile per i punti grafici: 0, 1, 2. Sfondo determinato dal 4	1017 byte

UNITÀ 21: GRAPHICS "Numero del modo grafico" + 16

Ogniqualvolta appare la finestra per il testo insieme alla finestra grafica sul fondo dello schermo sono disponibili quattro righe per il testo. Per poter usare l'intero video per la visualizzazione grafica, date il comando Graphics aggiungendo 16 al numero del modo grafico selezionato.

GRAPHICS 3



GRAPHICS 3 + 16



UNITÀ 22: Cambiamento dei modi grafici senza cancellazione del video.

GRAPHICS 1



GRAPHICS 2 + 32



UNITÀ 23: Esercitazione Pratica della Lezione Cinque

NOTA: Per cancellare quanto contenuto nella memoria RAM del computer, scrivere **NEW** e premere il tasto **RETURN** prima di iniziare ogni parte pratica e fra un programma esemplificativo e l'altro.

Eseguite questo programma

```
10 REM STAMPA A SCHERMO INTERO
20 GRAPHICS 2 + 16
30 FOR X= 1 TO 12
40 PRINT # 6; "LINEA"; X
50 NEXT X
80 FOR Y= 1 TO 1000:NEXT Y:REM RITARDO
90 END
```

Aggiungete ora le seguenti linee

```
60 FOR Y = 1 TO 1000:NEXT Y:REM RITARDO
70 GRAPHICS 1 + 16 + 32
```

NOTA: Il computer utilizza quantità di memoria diverse a seconda dei vari modi grafici. Se la memoria dello schermo fa apparire alcuni strani caratteri nella parte alta del video, quando viene eseguito questo programma, ciò è dovuto al cambiamento delle dimensioni della memoria utilizzata fra un modo di funzionamento grafico ed un altro.

Non appena pronti caricate la lezione successiva, scrivendo **CLOAD** e premendo due volte il tasto **RETURN**. Non appena il messaggio **READY** viene visualizzato sullo schermo, scrivete **RUN** e premete il tasto **RETURN** per iniziare la lezione.

LEZIONE SEI: POSIZIONAMENTO ED ANIMAZIONE

UNITÀ 24: Programma con POSITION

Il comando **PLOT** traccia i punti in una posizione precisa sul video. **POSITION** è simile, ma determina il punto di partenza dei testi che andranno posizionati sullo schermo.

Eseguite il programma seguente:

```
10 GRAPHICS 0
20 SETCOLOR 2, 14, 2
30 POSITION 12, 11
40 PRINT "METÀ"
```

NOTA: POSITION fissa solo il punto d'inizio del testo sullo schermo. **"PRINT"**, **"PRINT # 6"**; oppure **"DRAWTO"** devono precedere il testo da far apparire sul video.

UNITÀ 25: Istruzione POSITION

La sintassi dell'istruzione **POSITION** è:
POSITION X, Y dove X rappresenta il numero orizzontale ed Y rappresenta il numero verticale.

UNITÀ 26: Determinate l'esattezza dei programmi che seguono.

```
PROGRAMMA 1
10 PRINT "TESTO"
20 POSITION 5, 5
30 GRAPHICS 0
```

```
PROGRAMMA 2
10 GRAPHICS 2
20 POSITION 8, 4
30 PRINT # 6; "TESTO"
```

```
PROGRAMMA 3
10 GRAPHICS 1
20 POSITION "TESTO"
```

UNITÀ 27: Esempio di Animazione

```
READY
10 GRAPHICS 1
20 FOR X=1 TO 10
30 POSITION X,5
40 PRINT # 6; "PAROLA"
50 POSITION X, 5
60 PRINT # 6; "      "
70 NEXT X
```

UNITÀ 28: Esempio di Animazione

```
READY
10 GRAPHICS 1
20 FOR X=1 TO 10
30 POSITION X, 5
40 PRINT # 6; "PAROLA"
45 FOR TEMPO= 1 TO 20:NEXT TEMPO
50 POSITION X, 5
60 PRINT # 6; "      "
70 NEXT X
```

UNITÀ 29: Principi di Animazione del testo

1. Le parole devono essere posizionate e visualizzate sullo schermo
2. L'immagine deve essere percepita dall'occhio grazie ad un breve intervallo
3. L'immagine deve essere cancellata
4. Il testo riposizionato e visualizzato in una posizione leggermente diversa.

UNITÀ 30: Esercitazione pratica della Lezione Sei

NOTA: Per cancellare quanto contenuto nella memoria RAM del computer, scrivere **NEW** e premere il tasto **RETURN** prima di iniziare ogni parte pratica e fra un programma esemplificativo e l'altro.

Dimostrazione di POSITION in Graphics 0.

```
READY
10 GRAPHICS 0
20 PRINT "SCRIVI DUE NUMERI:
    IL PRIMO NUMERO È":PRINT
    "IL VALORE DI X, IL SECONDO DI Y."
30 PRINT: PRINT
40 INPUT X, Y
50 POSITION X, Y
60 PRINT "POSIZIONE: X = ";X;" ";Y=" ";Y
70 END
```

Dimostrazione di posizione casuale

```
READY
10 REM POSIZIONAMENTO CASUALE
20 GRAPHICS 2
30 POSITION 5,5
40 PRINT # 6; "GATTO"
50 X = INT(RND(0)*20):Y = INT(RND(0)*10)
60 POSITION X, Y:PRINT # 6;"0"
70 GOTO 30
```

Movimento di una freccia in orizzontale sul video

```
READY
10 GRAPHICS 2
20 FOR X = 1 TO 18
30 POSITION X,5
40 PRINT # 6;" —>"
50 FOR T = 1 TO 100:NEXT T
55 POSITION X, 5
60 PRINT # 6;" "
70 NEXT X
```

Movimento di una freccia in orizzontale senza un comando PRINT che elimini l'immagine. Fate partire questo programma e cercate di capire come l'immagine viene cancellata. La spiegazione verrà data in seguito.

```
READY
10 GRAPHICS 2
20 FOR X = 1 TO 17
30 POSITION X,5
40 PRINT # 6;" —> "
50 FOR T = 1 TO 100:NEXT T
70 NEXT X
```

Posizionando uno spazio bianco prima della freccia, si crea l'illusione del movimento ogni qualvolta viene visualizzata ad nuova posizione sul video.

Non appena pronti caricate la lezione successiva, scrivendo **CLOAD** e premendo due volte il tasto **RETURN**. Non appena il messaggio **READY** viene visualizzato sullo schermo, scrivete **RUN** e premete il tasto **RETURN** per iniziare la lezione.

LEZIONE SETTE: CARATTERI GRAFICI

UNITÀ 31: Istruzione POKE





Figura 4: Schema dei simboli grafici impostabili dalla tastiera.

Per posizionare i caratteri minuscoli e i caratteri grafici sul video nei modi di funzionamento grafico 1 o 2, usare l'istruzione **POKE**.

1. Selezionare un modo grafico (1,2)
2. Usare l'istruzione **POKE 756, 226**
3. PRINT # 6; "lettere o caratteri grafici"

UNITÀ 32: Programma con sfondo composto da Cuori.



UNITÀ 33: Programma con "sfondo pulito"

```
READY
10 GRAPHICS 1
20 POKE 756, 226
30 SETCOLOR 0, 0, 0
40 PRINT #6; "GRAPHICS 1 H _ _ / "
```

UNITÀ 34: Esercitazione pratica per la Lezione Sette

NOTA: Per cancellare quanto contenuto nella memoria RAM del computer, scrivere **NEW** e premere il tasto **RETURN** prima di iniziare ogni parte pratica e fra un programma esemplificativo e l'altro.

Inserire il seguente programma. I simboli grafici di questo programma, si ottengono in questo modo:

riga in alto = CTRL M, riga in basso = CTRL N
riga sinistra = CTRL V, riga destra CTRL B

```
READY
10 GRAPHICS 1
20 POKE 756, 226
30 SETCOLOR 0, 0, 0
40 PRINT #6; "  _____  "
50 PRINT #6; " |lettere | "
60 PRINT #6; " |minuscole| "
70 PRINT #6; "  _____  "
```

Eliminate la linea 20 del programma precedente ed eseguitelo
Reinserite la linea 20, eliminate la linea 30 e eseguitelo nuovamente.

Programma dimostrativo d'animazione

```
READY
10 GRAPHICS 1
20 POKE 756, 226
30 SETCOLOR 0, 0, 0
35 FOR X=0 TO 10
40 POSITION X, 5:PRINT # 6; " _____ "
50 POSITION X, 6:PRINT # 6; " |lettere |"
60 POSITION X, 7:PRINT # 6; " |minuscole|"
70 POSITION X, 8:PRINT # 6; " _____ "
80 FOR TEMPO = 1 TO 50:NEXT TEMPO
90 NEXT X
```

CONGRATULAZIONI, PASSATE AL QUIZ FINALE SULLA CASSETTA

Programmi dimostrativi

Il computer ATARI ha 15 modi grafici.


Di seguito troverete le rappresentazioni grafiche nei modi **6, 7, 8, 9, 10 e 11**.

Alcuni programmi di effetti grafici e sonori completano il corso di INVITO A PROGRAMMARE 3.

```
1 POKE 764,255
10 REM DISEGNO ANIMATO IN GRAFICA 3
1000 GRAPHICS 19:POKE 710,15
1020 X=1:XB=X:Y=1:YB=Y:DX=1:DY=1
1030 COLOR 1:PLOT 14,11:DRAWTO 26,11
1040 COLOR 1:PLOT 0,0:DRAWTO 38,0
1050 PLOT 0,23:DRAWTO 39,23
1060 COLOR 2:PLOT 0,1:DRAWTO 0,23
1070 PLOT 39,0:DRAWTO 39,22
1075 PLOT 20,6:DRAWTO 20,16
1080 COLOR 0
1090 PLOT XB,YB
1100 XB=X:YB=Y
1110 X=X+DX
1120 Y=Y+DY
1130 LOCATE X,Y,Z
1140 IF Z<>0 THEN GOTO 1190
1150 COLOR 3
1160 PLOT X,Y
1170 POKE 77,0
1180 GOTO 1080
1190 SOUND 0,100,10,10
1200 IF Z=1 THEN DY=-DY
1210 IF Z=2 THEN DX=-DX
1220 X=XB:Y=YB
1230 POKE 707+Z,PEEK(20)
1240 SOUND 0,0,0,0
1250 IF PEEK(764)=255 THEN GOTO 1110
```

Modo Grafico 6 e 7 con finestra del testo

52



```

10 REM CERCHIO IN GRAFICA 7+16
20 GRAPHICS 7+16:COLOR 1
30 FOR LOOP=1 TO 400
40 A=A+0.05
50 X=SIN(A)*40:Y=COS(A)*40
60 PLOT X+80,Y+45
70 NEXT LOOP
80 GOTO 80

```

```

1 POKE 764,255
2 C=1
3 GRAPHICS 7+16
5 E=INT(300*RND(1))
10 D=INT(300*RND(1))
15 C=1
20 COLOR C
40 B=39
42 A=79
45 FOR S=1 TO D STEP E
50 FOR X=A TO B STEP -2
60 PLOT 80,A-X
70 DRAWTO 80+X,INT(A/S)
80 DRAWTO 80,X
90 DRAWTO 80-X,INT(A/S)
95 DRAWTO 80,A-X
100 COLOR C
110 NEXT X
120 C=C+1
125 NEXT S
126 SETCOLOR 0,T,2
127 T=T+1
130 GOTO 5

```


```

1 POKE 764,255
10 REM CARATTERI IN GRAFICA 2
20 GRAPHICS 2+16
30 POKE 756,226
35 FOR J=1 TO 150
37 POSITION 0,0
40 FOR I=1 TO 80
50 ? #6;CHR$(J);CHR$(J+2);CHR$(J+2);
60 NEXT I
80 NEXT J

```

Modo Grafico 8 con finestra del testo

54



```

10 REM CERCHIO IN GRAFICA 8+16
20 GRAPHICS 8+16:POKE 709,14:COLOR 1
30 FOR LOOP=1 TO 400
40 A=A+0.05
50 X=SIN(A)*50:Y=COS(A)*50
60 PLOT X+160,Y+96
70 NEXT LOOP
80 GOTO 80

```

```

10 REM ELLISSE IN GRAFICA 8+16
20 GRAPHICS 8+16:POKE 709,14:COLOR 1
30 FOR LOOP=1 TO 400
40 A=A+0.05
50 X=SIN(A)*80:Y=COS(A)*15
60 PLOT X+160,Y+96
70 NEXT LOOP
80 GOTO 80

```

```

5 REM CURVE IN GRAFICA 8
10 READ Y:IF Y=0 THEN Y=RND(1)*5+1:RESTORE 999
20 GRAPHICS 8:SETCOLOR 1,1,14:SETCOLOR 2,0,0:COLOR 1:PLOT 0,80
30 FOR X=0 TO 319:DRAWTO X,80+INT(79*SIN(X*Y)):NEXT X
35 ? "PREMI UN TASTO PER UN NUOVO DISEGNO"
40 P=PEEK(764):POKE 764,255:IF P=255 THEN 40
50 GOTO 10
995 DATA 1.55111,3.1,4.13186,.3,2.34573,.6
999 DATA 0

```

```


10 REM STELLE IN GRAFICA 8
15 DEG :R=50:COLOR 1
20 GRAPHICS 8:SETCOLOR 4,14,0:SETCOLOR 2,12,2
25 PLOT R+160,80
30 FOR T=0 TO 1800 STEP 160
40 DRAWTO 160+R*COS(T),80+R*SIN(T)
50 NEXT T
60 PRINT "SCRIVI UN NUMERO COMPRESO TRA 10 E 80";:INPUT R
70 GOTO 25

```

NOTA: SOSTITUIRE LA LINEA 30 IN:
 30 FOR T=0 TO 1800 STEP 100

Modo Grafico 6 e 7

56



```

10 REM SENO E COSENO
20 REM IN GRAFICA 7
30 GRAPHICS 7
40 FOR A=0 TO 10 STEP 0.1
50 COLOR 2
60 PLOT A*15,SIN(A)*20+40
70 COLOR 1
80 PLOT A*15,COS(A)*30+40
90 NEXT A

```

```

5 PRINT CHR$(125)
10 PRINT "SCRIVI UN NUMERO COMPRESO TRA 1 E 800":INPUT NUMERO
20 FOR EX=1 TO 3
30 GRAPHICS 7:POKE 710,0
40 COLOR 1
50 PLOT 2,2:DRAWTO 158,2
60 DRAWTO 158,80
70 DRAWTO 2,2:POKE 752,1:PRINT "RITARDO IN GRAFICA 7"
80 FOR RITARDO=1 TO NUMERO:NEXT RITARDO
90 GRAPHICS 0:POKE 752,1
100 SETCOLOR 2,8,6:SETCOLOR 1,0,12
110 POSITION 8,11
120 PRINT "RITARDO IN GRAFICA 0"
130 FOR RITARDO=1 TO NUMERO:NEXT RITARDO
140 RUN

```


```

5 REM CALEIDOSCOPIO IN GRAFICA 3
10 GRAPHICS 3
20 X=INT(RND(1)*20):Y=INT(RND(1)*20)
30 COLOR INT(RND(1)*4)
40 PLOT X,Y:PLOT 19-X,Y:PLOT X,19-Y:PLOT 19-X,19-Y:PLOT 20+X,Y:PLOT 39-X,Y:PLOT
20+X,19-Y:PLOT 39-X,19-Y
41 PLOT Y,X:PLOT 19-Y,X:PLOT Y,19-X:PLOT 19-Y,19-X:PLOT 20+Y,X:PLOT 39-Y,X:PLOT
20+Y,19-X:PLOT 39-Y,19-X
60 GOTO 20

```

Modo Grafico 8

58




```
5 REM COLORI IN GRAFICA 11
10 GRAPHICS 11
20 FOR Z=0 TO 120
30 COLOR Z
40 PLOT 0,Z:DRAWTO 79,Z
50 NEXT Z
60 GOTO 80
```


```
5 REM BARRE DI COLORE IN GRAFICA 9
10 GRAPHICS 9
20 SETCOLOR 4,4,0
30 COLOR 12
40 PLOT 39,2:DRAWTO 39,191
50 PLOT 40,2:DRAWTO 40,191
60 COLOR 4
70 PLOT 41,2:DRAWTO 41,191
80 PLOT 42,2:DRAWTO 42,191
90 GOTO 90
```

```
5 REM COLORI IN GRAFICA 9
10 GRAPHICS 9
20 FOR Z=64 TO 80
30 COLOR Z
40 PLOT 2,Z:DRAWTO 79,Z
50 NEXT Z
60 FOR K=95 TO 80 STEP -1
70 READ X:COLOR X
80 PLOT 2,K:DRAWTO 79,K
90 NEXT K
100 FOR FL=0 TO 15:SETCOLOR 4,FL,0:FOR DE=1 TO 300:NEXT DE:NEXT FL:GOTO 100
110 DATA 0,1,2,3,4,5,6,7,8,9,10,11,12,13,14,15
```

```
5 REM COLORI IN GRAFICA 10
10 GRAPHICS 10
20 FOR X=15 TO 85
30 N=INT(RND(0)*8)+1
40 COLOR N
50 PLOT X,2:DRAWTO X,191
60 NEXT X
80 FOR COLORE=704 TO 712
90 K=INT(RND(0)*255)
100 POKE COLORE,K
110 NEXT COLORE
120 FOR RIT=1 TO 100:NEXT RIT
130 GOTO 80
```

Modo Grafico 9, 10 e 11

09



```

1 REM ASTRONAVE IN GRAFICA MOSSA DAL JOYSTICK 1
5 POKE 752,1:PRINT CHR$(125)
10 SETCOLOR 2,0,0:X=120:Y=48
20 A=PEEK(106)-8:POKE 54279,A:PMBASE=256*A
30 POKE 559,46:POKE 53277,3
40 POKE 53248,X
50 FOR I=PMBASE+512 TO PMBASE+640:POKE I,0:NEXT I
60 POKE 704,216
70 FOR I=PMBASE+512+Y TO PMBASE+516+Y:READ A:POKE I,A:NEXT I
80 DATA 153,189,255,189,153
100 A=STICK(0):IF A=15 THEN GOTO 100
110 IF A=11 THEN X=X-1:POKE 53248,X
120 IF A=7 THEN X=X+1:POKE 53248,X
130 IF A=13 THEN FOR I=6 TO 0 STEP -1:POKE PMBASE+512+Y+I,PEEK(PMBASE+511+Y+I):N
EXT I:Y=Y+1
140 IF A=14 THEN FOR I=0 TO 6:POKE PMBASE+511+Y+I,PEEK(PMBASE+512+Y+I):NEXT I:Y=
Y-1
150 GOTO 100

```

```

5 REM SCHERMO COLORATO CON DUE COLORI
10 A=PEEK(560)+256*PEEK(561)+15
20 POKE A,130
30 FOR I=1 TO 16:READ A:POKE 1536+I,A:NEXT I
40 DATA 72,169,80,141,10,212,141,23,208,169,88,141,24,208,104,64
50 POKE 512,1:POKE 513,6
60 POKE 54286,192

```

```

5 REM CILINDRI IN GRAFICA 9
10 GRAPHICS 9:SETCOLOR 4,5,0
20 FOR Y=55 TO 0 STEP -10
30 FOR X=0 TO 24
40 C=X:IF X>11 THEN C=24-X
45 C=C+3
50 Z=Y+(X)
55 D=INT(SQR(144-(X-12)*(X-12)))/2
57 COLOR 15-C
58 PLOT Z,Y+7-D
60 DRAWTO Z,Y+7+D
70 COLOR C
80 DRAWTO Z,180-Y+D
180 NEXT X
190 NEXT Y
200 GOTO 200

```

```

10 REM COLORI ANIMATI IN GRAFICA 10
110 DIM C(8):GRAPHICS 10
115 FOR Z=704 TO 712:READ R:R=R*16+8:C(Z-704)=R:POKE Z,R:NEXT Z
116 DATA -.5,1,3,4,5,7,9,12,13
118 LIM=22:T2=3.14159*2/LIM:COL=3:E1=1:DIM D(LIM,2)
120 GOSUB 1500:FOR V=1 TO LIM:T=T+T2:GOSUB 1500:NEXT V
400 GOTO 1000
490 REG=705
500 FOR X=1 TO 8:POKE REG,C(X):REG=REG+1:IF REG>712 THEN REG=705
510 NEXT X:REG=REG+1:IF REG>712 THEN REG=705
520 POKE 77,0:GOTO 500
1000 REM
1005 FOR E=1 TO 10:E2=INT(E/2-0.5)
1010 FOR R=E1 TO E1+E2:CR=8-COL:IF CR=0 THEN CR=8
1015 V=0:COLOR CR:GOSUB 2000:PLOT X,Y
1020 FOR V=1 TO LIM:T=T+T2:GOSUB 2000:DRAWTO X,Y:IF V>=LIM/2 THEN COLOR COL
1025 NEXT V:NEXT R:COL=COL+1:IF COL=9 THEN COL=1
1030 E1=E1+INT(E/2+0.5):NEXT E
1200 GOTO 490
1500 D(V,1)=SIN(T):D(V,2)=COS(T):RETURN
2000 X=(30-R)*0.6*D(V,1)+40:Y=60*D(V,2)+80:RETURN

```

```

5 REM COLORI E GRAFICA 10
10 GRAPHICS 10
20 FOR A=705 TO 712:POKE A,INT(RND(0)*255):NEXT A
30 COLOR INT(RND(0)*7)+1:PLOT INT(RND(0)*79),INT(RND(0)*191):DRAWTO 39,95
40 GOTO 20

```


```

1 REM DISEGNO IN GRAFICA 11
5 GRAPHICS 11
10 FOR Z=0 TO 79
20 PLOT 0,Z
30 COLOR Z
40 DRAWTO Z,79
50 NEXT Z
60 GOTO 60
70 PLOT Y,Z
80 COLOR Y
90 DRAWTO Y,79
100 NEXT Y
110 GOTO 10

```

Indice

1. INTRODUZIONE	pag.	1
2. SPIEGAZIONE DEL CORSO	»	2
Componenti Atari	»	2
Istruzioni di carattere generale per il caricamento del programma (o car. da nastro)	»	2
Preparazione per le esercitazioni pratiche	»	4
3. RIASSUNTO DELLE LEZIONI: INTRODUZIONE AL SUONO	»	6
4. DESCRIZIONE DELLE UNITÀ PER INTRODUZIONE AL SUONO ...	»	8
Lezione Uno: Registri Sonori	»	8
Lezione Due: Effetti Sonori	»	9
Lezione Tre: Routine Sonore	»	12
Lezione Quattro: Programmi Sonori	»	16
Lezione Cinque: ABC Musicale	»	20
Lezione Sei: Riconoscimento delle Note	»	22
Lezione Sette: Routine Sonore con Registri Multipli	»	26
Quiz Finale: INTRODUZIONE AL SUONO	»	30
5. RIASSUNTO DELLE LEZIONI: INTRODUZIONE ALLA GRAFICA ...	»	31
6. DESCRIZIONE DELLE UNITÀ PER L'INTRODUZIONE ALLA GRAFICA ...	»	32
Lezione Uno: Dimostrazione Grafica	»	32
Lezione Due: Uso dei colori	»	32
Lezione Tre: Modi di Funzionamento Grafico 1 e 2	»	34
Lezione Quattro: Grafica	»	38
Lezione Cinque: Trucchi con Modi di Funzionamento Grafico	»	41
Lezione Sei: Posizione ed Animazione	»	44
Lezione Sette: Caratteri Grafici	»	48
Quiz Finale: INTRODUZIONE ALLA GRAFICA	»	50
Ulteriori informazioni:	»	51



ATARI[®] ITALIA S.p.A

Via dei Lavoratori, 19 • 20092 Cinisello B. (MI)